# QMAKE Configuration file
# Copyright (c) 2007-2010, WOSH - Wide Open Smart Home 
# by Alessandro Polo - OpenSmartHome.com
# All rights reserved.
#######################################################
# Revision $Id: woshkiosk.pro 2275 2010-04-06 02:00:33Z alex $
#######################################################

message( "WOSH Kiosk (multi-platform)" )

#####################################################
# generic configuration options
# woshkiosk is a kiosk application (styled-gui)
#
TEMPLATE = app
TARGET = woshkiosk


#contains(QT_CONFIG, opengl):QT += opengl
QT += opengl

#####################################################
# set namespaces of the WOSH Framework to build-in
#
CONFIG += WOSH_AUTOMATIONS
CONFIG += WOSH_BUILDING
CONFIG += WOSH_COMMUNICATION
CONFIG += WOSH_ENTERTAINMENT
CONFIG += WOSH_PLUGINS
CONFIG += WOSH_SECURITY
CONFIG += WOSH_XTENDED

CONFIG += WOSH_GUI
CONFIG += WOSH_KIOSK


#####################################################
# include core and interfaces, see
# framework/framework.pro for options, many
# namespaces are disabled by default
#
!include( common.pri ) {
	error( "FATAL: Project file 'common.pri' is missing! (misaligned distribution?)" )
}
#####################################################
# include WOSH Framework, framework options must be
# configured before following command.
#
!include( framework/framework.pro ) {
	error( "FATAL: Project file 'framework/framework.pro' is missing! (misaligned distribution?)" )
}
#####################################################
# force inclusion of DiscoveryUdp service
#
!include( bundles/services/DiscoveryUdp/DiscoveryUdp.prf ) {
	message( "*****  No bundles/services/DiscoveryUdp/DiscoveryUdp.prf file found. SKIPPED. *****" )
} 
#####################################################
# define application's resources
#
RESOURCES += apps/woshkiosk/woshkiosk.qrc
RESOURCES += apps/woshkiosk/woshkiosk_styles.qrc

#####################################################
# include application's core file
#
SOURCES += apps/woshkiosk/woshkiosk.cpp

HEADERS += apps/woshkiosk/WoshKioskWindow.h
SOURCES += apps/woshkiosk/WoshKioskWindow.cpp


